* clean_prices.do
* This file cleans the prices data to be merged to the labor datasets
* Inputs : Data/Importable/Precios/precios98completo.dta
* Outputs: Data/Clean/Precios/precios98completo_clean.dta
* This is based on the old file base99 in the old directory

clear all
version 13.1

* Project info
project, doinfo   
local master "`r(pdir)'"
local doname "`r(dofile)'" 

* cd "C:/Users/JorgeEduardo/Dropbox/AA_Minimum_Wage"
project, original("`master'/Data/Importable/Precios/precios98completo.dta")
use "`master'/Data/Importable/Precios/precios98completo.dta", clear
* No necesito variacion
drop variacion*
* Quita los espacios
replace grupo=subinstr(grupo," ","",.)
keep if grupo=="Aniveltotal"
replace estrato=subinstr(estrato," ","",.)
* Cambio la forma de la base
reshape wide valor_indice, i(ano mes ciudad estrato) j(grupo) string
ren valor ipc
ren ano year
gen trimestre=.
replace trimestre=1 if inlist(mes,1,2,3)
replace trimestre=2 if inlist(mes,4,5,6)
replace trimestre=3 if inlist(mes,7,8,9)
replace trimestre=4 if inlist(mes,10,11,12)
* Llenar los missings del estrato alto, una regresi�n com�n y corriente
reshape wide ipc, i ( year mes ciudad) j(estrato) string
xi: reg  ipcAlto ipcAniveltotal ipcBajo ipcMedio i.ciudad i.mes i.year
predict y
replace ipcAlto=y if ipcAlto==.
drop y
* Para el an�lisis inicial solo necesito estos a�os, cambiar si llego a necesitar m�s a�os.
* N�tese que estoy botando los a�os extra despu�s de hacer la regresi�n. Us� los a�os que no necesitaba para imputar, as� tuve m�s informaci�n.
keep if year>=1996 & year<=2000
reshape long 
collapse (mean) ipc , by(trimestre year ciudad estrato)
replace ciudad=upper(ciudad)
replace ciudad="BOGOT�" if ciudad=="BOGOTA"
replace ciudad="MEDELL�N" if ciudad=="MEDELLIN"
encode estrato, gen(est)
drop estrato
tempfile estrato_ciudad estrato_ciudad_serie total_ciudad  total_ciudad_serie total_pais total_pais_serie
* Estrato - ciudad
preserve
drop if est==2
ren ipc ipc_ec
* keep if est==2
save `estrato_ciudad'
egen j=concat(year trimestre)
destring j, replace
drop year trimestre
reshape wide ipc_ec, i(ciudad est) j(j)
* reshape wide ipc, i(ciudad) j(j)
save `estrato_ciudad_serie'
use `estrato_ciudad'
merge n:1 ciudad est using `estrato_ciudad_serie'
drop _merge
ren trimestre trim
save "`master'/Data/Clean/Precios/precios98_ec.dta", replace
restore
* Ciudad
preserve
keep if est==2
ren ipc ipc_c
save `total_ciudad'
egen j=concat(year trimestre)
destring j, replace
drop year trimestre
reshape wide ipc_c, i(ciudad est) j(j)
* reshape wide ipc, i(ciudad) j(j)
save `total_ciudad_serie'
use `total_ciudad'
merge n:1 ciudad est using `total_ciudad_serie'
drop _merge
ren trimestre trim
save "`master'/Data/Clean/Precios/precios98_c.dta", replace
restore
* Pa�s
preserve
keep if est==2
keep if ciudad=="A NIVEL NACIONAL"
ren ipc ipc_p
save `total_pais'
egen j=concat(year trimestre)
destring j, replace
drop year trimestre
reshape wide ipc_p, i(ciudad est) j(j)
* reshape wide ipc, i(ciudad) j(j)
save `total_pais_serie'
use `total_pais'
merge n:1 ciudad est using `total_pais_serie'
drop _merge
ren trimestre trim
save "`master'/Data/Clean/Precios/precios98_p.dta", replace
restore



* Project calls
project, creates("`master'/Data/Clean/Precios/precios98_ec.dta")
project, creates("`master'/Data/Clean/Precios/precios98_c.dta")
project, creates("`master'/Data/Clean/Precios/precios98_p.dta")
